
Main References

Excel File and Slides
B 03_LogReturns.xlsx

m Lesson page, with the slides and the workbook download:

https://gianlucafusai.com/excelling-in-finance/computing-returns
Lecture Notes

m Fusai, G., Alberici, L. L., Migliardi, M. “Log-Returns in Excel 365: From Price Arrays to Return
Arrays with Dynamic Arrays and LAMBDA.” Excelling in Finance tutorial series.

Related Material
m  Microsoft documentation: DROP, TAKE, INDEX, SEQUENCE, TRIMRANGE, LAMBDA, LET, ISOMITTED.

m Campbell, Lo, MacKinlay, The Econometrics of Financial Markets, Chap. 1 (returns,
aggregation).

Agenda

m The input: a named price column (trim operator :.)
m Log-refurns vs. simple returns (one-page recap)

m The core idea: shifted-array alignment

® Main path: one-period returns via DROP

m One rule, seven formulas

m k-period returns: overlapping vs. non-overlapping

® Where the remainder goes: end vs. start indices

m Reusable LAMBDA functions: £x_LogReturn, fx_Return

m  Workflow summary, exercises, bridge to the next tutorial

Prices In, Returns Out



The Input: a Named Price Column

m Type or paste the closing prices in one column, oldest on top. Any source works: a
download, a data provider.

m Name the column in Formulas > Name Manager with the trim-range operator : ., and count

the prices:
d_HistPrices ='LogReturns (1 period) '!$C$9:.$C$1000
n_NObs =ROWS (d_HistPrices)

B C9:.C1000 is C9:C1000 without its trailing empty cells: the same as TRIMRANGE (C9:C1000, 2).
m The name always ends at the last price, and grows when prices are added below.

m Prefixes: d- dynamic range (changes size), n- any other name (inputs, constants), £x_
LAMBDA function.

m Requires Excel 365 or Excel for the web.

Fact

Pitfall

Keep the column empty below the prices: anything typed there joins the range, and every retumn
computed from it.

From here on every formula uses only d_HistPrices and n_NObs.

Prices In, Returns Out

Old world:  =LN(B2/B1) in a cell, drag a thousand rows.

Excel 365: one formula, the entire return vector; its length follows the price column automatically.

Every formula in this deck is one of:

Pattern Excel operator Role here

Named input trim operator : ., ROWS Input price array and its length
Dynamic reference DROP, TAKE, INDEX Aligned shiffed views

Pointwise fransform LN, /, —1 Ratio + logarithm, elementwise

Fact

No recursion here

Unlike the ScAN tutorial, nothing in this deck is recursive. A return depends on two prices at the same
time index, not on the previous return. Pointwise array formulas, not SCAN, are the right tool.

Log and Simple Returns in
One Slide

Log vs. Simple Returns

With P(t) the price at time t and A the sampling step:

Ra(t) = % _

ra(f) = M%) =In(1+Ra(1). ra(t) € (—o0,00).

1, Ra(t) € (—1.00)

Why finance prefers log-returns:
m Additivity. r(t,t + kA) = j’:o‘ ra(t +jA) by telescoping. Simple returns compound
multiplicatively.
m Support. Unbounded below; simple returns are capped at —100%.

m Symmetry. r and —r cancel exactly; —10%/+10% simple does not restore the investment.

Fact

One-line takeaway

Sums are easier to model than products.

Small-move approximation: r = R — R?/2; the two measures coincide for small moves
and diverge for large ones.

The Core Idea



Shifted-Array Alignment

This is the only idea of the whole deck.

A return is a pointwise tfransformation of the element-by-element ratio of two aligned price arrays:
m Numerator: later prices (Py, Ps. ... ‘Pn)T
m Denominator: earlier prices (Py, Py, ..., Pr_y )-r

m Both have length n — 1: the original array, each missing one endpoint.

Fact

The one rule of return computation

numerator = drop the first price
denominator = drop the last price
return = LN(numerator / denominator)

Every implementation below realises the two drops with a different Excel operator.

Alignment, made concrete

The five prices of the sheet LogReturns (1 period): (97.4, 99.2, 94.2, 98.5, 101.8) in €9:C13.

i 1 2 3 4

numerator DROP (d_HistPrices, 1) 99.2 94.2 98.5 101.8
denominator DROP(d_HistPrices, -1) 974 99.2 94.2 98.5

R =Puq/P—1 +1.85% —5.04% +4.56% +3.35%
1 =1In(Pii1 /P) 11.83% —5.17% +4.46% +3.30%

Fact

What to notice

Two DROPs line the array up against its own shifted copy. Pointwise ratio — gross returns. Apply LN (or
—1) elementwise. No copy-down, no recursion, no size known in advance.

Main Path: DROP



One-Period Returns with DROP

Log-returns: one formula (cell E10).

=LET(Prices_, d_HistPrices,

LN (DROP(Prices_, 1) / DROP(Prices_, -1)))

Simple returns: replace LN with a frailing —1 (cell F10).

=LET(Prices_, d_HistPrices,
DROP (Prices_, 1) / DROP(Prices_, -1) - 1)

Spills info a column of length n — 1: four returns from five prices, the first log-return being 1.83%.

Fact

Three reasons to prefer DROP

m No length counter. Unlike TAKE, INDEX, OFFSET, DROP does not need to know n: add a price

below the column and the returns follow.

m Non-voldtile. OFFSET recalculates on every workbook change; DROP does not.

m Readable. DROP(d_HistPrices, 1) / DROP(d_HistPrices, -1) reads like the shift rule itself.

One Rule, Seven Formulas



Seven Formulas, Same Result

Row 10 of LogReturns (1 period) writes the rule with seven formulas (E-K) plus £x_LogReturn (L),
each starting with =LET (Prices_, d-HistPrices, ...). Numerator on top, denominator below;
LN(numerator / denominator) gives the log-refurns.

Col. Operator Numerator / Denominator

E DROP (main path)  DROP(Prices-, 1)
DROP(Prices., -1)

G INDEX (range form) INDEX(Prices_,2,1):INDEX(Prices_,n_NObs,1)
INDEX(Prices-,1,1) :INDEX(Prices_,n_-NObs-1,1)

H INDEX+SEQUENCE  INDEX(Prices., SEQUENCE(nNObs-1, 1, 2))
INDEX (Prices., SEQUENCE(n_NObs-1))

I  OFFSET (volatile) OFFSET(Prices_, 1, 0, n.NObs-1, 1)
OFFSET(Prices., 0, O, n.NObs-1, 1)

J  TAKE TAKE (Prices._, 1-n_NObs)
TAKE (Prices_, n-NObs-1)

Column F is the simple return. Row 16: largest difference of each log-return column from column E
—allo.

A LAMBDA in a Cell, then a Named Function

Column X: an anonymous LAMBDA, defined and called in the same cell.

=LAMBDA (Prices_,
LET (num_, DROP(Prices_, 1), den_, DROP(Prices_, -1),
LN(num_ / den_)))(d_HistPrices)

® Prices_is the parameter, as in £x_LogReturn; the trailing (d_HistPrices) calls the function.

m Same values as column E: the point is syntactic.

Column L: the same idea, saved in the Name Manager as £x_LogReturn.

=fx_LogReturn(d_HistPrices)

Fact

From formula to function

Write the LAMBDA in a cell, test it on real data, then save it under a name. The sheet prints each
formula with FORMULATEXT just above its column, in row 8.

Pros and Cons

Operator Readable Needsn? Volatile?
DROP yes no no
TAKE yes yes no
INDEX+SEQUENCE no yes no
INDEX (range form) no yes no
OFFSET yes yes yes (legacy)

Fact

The default choice

DROP wins on all three criteria simultfaneously. Use DROP unless a constraint forces otherwise; avoid
OFFSET in analytics because of volafility.

k-Period Returns



From One Period to k Periods

k-period log-refurn:
k—1

1) = M%) = S ra(t+j8).
j=0

Given n prices, two sampling conventions:

Convention Count Windows

Overlapping n—k Mechanical overlap induces serial dependence
Non-overlapping m = [(n — 1)/k] No mechanical overlap between windows

Fact

When to use which
Overlapping: descriptive statistics, volatility scaling, large-sample estimators.

Non-overlapping: hypothesis tests, i.i.d.-based models, clean backtests.

Toy Example: n =11, k=3

Scheme Start (denom.) End (num.) Count

Overlapping 1.2,...8 4,5...11 n—k=28

Non-overlapping (anchored at the latest price) 2,5,8 5,8,11 [10/3] =3

end indices 5,8, 11
(fx-Return)

startindices 1,4,7
g=1 (alternative, not used)

m Endindices: SORT(SEQUENCE(3, 1, 11, -3)) =5,8,11;start=end —k =2,5,8.

m The last return ends on Pyy; the remainder g = MoD(10, 3) = 1 period (P, — P,) is unused, at
the start.

Overlapping Windows in the Workbook

Sheet LogReturns (k period Overlap): 22 pricesin G9:G30, k = 6 in D7 (hame n_k). One formula,
generalising the one-period case:

=LET(Prices_, d_HistPrices,
LN (DROP(Prices_, n_k) / DROP(Prices_, -n_k)))

The sheet splits it info columns, fo inspect each step:

Cell Formula Content

H9  =SEQUENCE(n_NObs-n_k) Start: 1,....16

I9 =H9#+n k End = Start + k: 7,...,22
J9  =DROP(d.HistPrices,n.k) Nurmerator

K9 =DROP(d_HistPrices,-nk) Denominator

L9 =LN(J9#/K9#) Log-return

M9 =fx Return(d_HistPrices,nk) Same, with the function

Fact

Result

n—k = 16 returns; the first is In(49.00/46.99) = 4.19%. D11 = largest difference between L and M = 0.

Non-Overlapping Blocks in the Workbook

Sheet LogReturns (k period not Overl): the same 22 prices, k = 6.

Cell Formula Content

D9  =ROUNDDOWN ((n_NObs-1)/n_k,0) m = 3 returns (hamMe n_m)

D10 =MOD(n_NObs-1,nk) g = 3 leftover periods

I9 =SORT(SEQUENCE(n.m,1,n.NObs,-nk)) Endindex: 10, 16,22

J9  =I9%-nk Start index: 4,10, 16

K9  =INDEX(d-HistPrices,I9#) Numerator

L9 =INDEX(d-HistPrices,J9#) Denominator

M9 =LN(K9#/L9#) First: In(45.60/47.92) = —4.96%

N9 =fx Return(d_HistPrices,n_k,FALSE) Same, with the function

Fact

Anchored at the latest price

SEQUENCE starfs at n = 22 and steps back by k; SORT puts the end indices in date order. Blocks
Py — P1o. Pio = Pis. P16 — Pao: the last return ends on the last price, and the g = 3 periods from P,
to P4 (grey in the sheet) are unused.

Where Does the Remainder Go? End vs. Start Indices

n prices give n — 1 one-period intervals:

n-l=mk+q, m=|"]. g=(-Nmedk 0<qg<k

End indices (fx_Return) Start indices (alternative)
Anchor Last price, steps backward First price, steps forward
Indices SORT (SEQUENCE (nm, 1,n_NObs,-n_k)) SEQUENCE(nm,1,1,nk)
Remainder At the start At the end
Pn used Always Onlyifg=0
Py used Onlyifg=0 Always
n=22,k=6 Py — Pig — P1g = Py Py — Py — P13 — Pyg

The two coincide when MoD(n — 1, k) = 0, and always for overlapping windows.

Fact

Neither is more correct

End-anchored when the series must end with the latest observation; start-anchored when a fixed
start date matters. Choose consciously, and name the function so the choice stays visible.

Reusable L.AMBDA Functions



fx_LogReturn: the Minimal Wrapper

Defined once in the Name Manager; called anywhere in the workbook as
=fx_LogReturn(d_HistPrices).

most recent on the bottom, normally the named
range d_HistPrices (see "The price series").

=LAMBDA (Prices_,
LET(
Textl_, "Prices_: ome column, oldest price on top",
Text2_, "Shift rule: drop the first price (numerator), drop the last price (denominator)",
num_, DROP(Prices_, 1),
den_, DROP(Prices_, -1),
Text3_, "One-period log-returns: n - 1 values, in date order",

B num_ = DROP(Prices_, 1) is the numerator, den. = DROP(Prices_, -1) the denominator; LN (num_
/ den.) applies pointwise. It is the shift rule.

Fact

Naming convention

Parameters and LET variables end with an underscore (Prices., num_, den_), so they never clash with
cell addresses or worklbook names (d_-HistPrices, n k). Text1_, Text2_, ... Are comments: LET
accepts names that are never used.

fx Return(Prices_, [k.], [overlap.], [log-]1): the Code

One function for every case: k periods, overlapping or not, log or simple. Defaults: k = 1,

overlapping, log.

1/0 (in Excel TRUE = 1 is FALSE, so IF(overlap = 1, ...) ignored

TRUE) .

=LAMBDA(Prices_, [k_1, [overlap_l, [log.l,

LET(
Textl_, "Defaults: one period, overlapping windows, log-returns (flags accept 1/0 or TRUE/FALSE)",
h_, IF(ISOMITTED(k), 1, k),
ov_, IF (ISOMITTED (overlap_), TRUE, overlap_),
lg_, IF(ISOMITTED(log_), TRUE, log.),
Text2_, "Non-overlapping blocks end at n, n-k, n-2k, ...: anchored at the latest price, sorted into date order",
n_, ROWS (Prices_),
endIx_, SORT(SEQUENCE(ROUNDDOWN((n_ - 1) / h_, 0), 1, n_, -h_)),
Text3_, "Shift rule: numerator = later prices, denominator = prices k periods earlier",
num_, IF(ov_, DROP(Prices_, h_), INDEX(Prices_, endIx_, 1)),
den_, IF(ov_, DROP(Prices_, -h_), INDEX(Prices_, endIx_ - h_, 1)),
Text4_, "Gross return, then log-return or simple return",
gross_, num_ / den_,
overlap.  log-  Output (in date order)
1/TRUE 1/ TRUE Overlapping k-period log-returns (default)
1/ TRUE O/ FALSE Overlapping k-period simple returns
0/ FALSE 1/ TRUE Non-overlapping k-period log-returns
0/ FALSE 0/ FALSE Non-overlapping k-period simple retumns

fx_Return: How It Works

m Optional arguments. [k_]1, [overlap_], [log-] may be omitted; ISOMITTED supplies the
defaults h- = 1, ov_ = TRUE, 1g_ = TRUE.

m Overlapping. DROP(Prices_, h_) / DROP(Prices_, -h_): n — K returns.

m Non-overlapping. endIx. = end indices n,n — k,n — 2k, ..., sorted into date order; INDEX
reads the prices at endIx_and endIx_ - h_: ROUNDDOWN((n. - 1) / h_, 0) refurns, the last one
ending on the last price.

m Flags. IF(ov_, ...), IF(1g-, ...): TRUE or any non-zero number takes the first branch, so 1/0
and TRUE/FALSE both work.

m Return type. gross. = num_ / den_, then LN(gross_) or gross_ - 1.

fx_Return: Typical Calls and Checks

= fx_Return(d_HistPrices) -- one period, log (= fx_LogReturn)

= fx_Return(d_HistPrices, 5) -- 5 periods, overlapping, log

= fx_Return(d_HistPrices, 5, FALSE) -- 5 periods, non-overlapping, log

= fx_Return(d_HistPrices, 21, 0, 0) -- 21 periods, non-overlapping, simple

Sheet £x_Return (k = 5): the four cases in columns I-L, your choice in H (n_k, n_overlap, n_log), and
four checks, all 0:

B oOne period: fx_Return = fx_LogReturn;
m felescoping: sum of the log-returns = In(P, /P):
B In(1+R)=r;

® TRUE/FALSE = 1/0.

Fact

One formula, the entire pipeline

Every downstream formula in the workbook becomes a single £x_Return(...) call. The sampling
logic is centralised in one place; any change propagates automatically.

28



Workflow and Bridge

29



The Five-Step Pipeline

1. Paste the prices one column, oldest on top

|

2. Name the column

!

[ J
[ J
[ 3. Shift-align via DROP } DROP(d_HistPrices, +1)
[ J
[ J

d_HistPrices = C9:.C1000
n_NObs

|

4. Pointwise transform (LN / —1)

!

5. Package as LAMBDA

LN (numerator / denominator)

fz_LogReturn, fr_Return

Every formula in this tutorial is a step in this single pipeline.

The Central Message

Fact

A return is an aligned ratio of two shifted price arrays. Dynamic arrays — and DROP in particular — let
Excel carry out that alignment in one formula.

Two takeaways:
m Classical spreadsheets ask the analyst to enumerate the rows. Dynamic arrays ask the
analyst to describe the array. The descriptive perspective is shorter and more robust.
m Nothing in this deck is recursive. Compare with the ScaAN tutorial, where the running statistics
M;, MDDy, etc. are recursive. Knowing which side of the divide a new formula lives on is the
single most transferable mental reflex.

Exercises

All on 03_LogReturns.x1sx.

1.

Telescoping. Check that =suM(E10#) equals In(101.8/97.4) = 4.42%; repeat with the simple
returns in column F and explain the gap.

The trimmed range. Type a sixth price in ¢14: does every column of row 10 gain a return?
Then type a note in ¢20: explain the errors.

Changing k. On the non-overlapping sheet set n_k to 7, then to 5. How many returns, how
many grey periods? Check against m and q.

End vs. start indices. Build the start-anchored version with SEQUENCE (nm,1,1,n k) and
compare it with fx_Return(d-HistPrices,n_k,FALSE). When do they agree?

Voldtility scaling. Annualise the standard deviation of one-period log-returns with v/252, /52
or V12. Why is the vk rule exact only for uncorrelated returns?

Next Steps

Next tutorial: Descriptive Statistics of Financial Returns.
The return array produced here is the direct input to:
® sample mean, variance, skewness, kurtosis;
m autocorrelation and Ljung-Box test;
m histogram, QQ-plot, kernel density;

m stationarity diagnostics.

Further down the series:
m Covariance matrix estimation, GMV portfolio, efficient frontier.
m EWMA and GARCH volatility modelling (recursive again, via SCAN).

® Brownian-motion simulation and option pricing.
The scan-based running statistics of the SCAN tuforial and the

pointwise return construction of this one are the two basic
building blocks on which the rest of the series rests.

Thank You

gianluca.fusai@uniupo.it
gianluca.fusai.1l@citystgeorges.ac.uk

